
第4回　メモリの導入

データメモリの取り付け

前回設計したデータパスでは、アキュムレータを使うことで、演算した結果に対して次々に演算を施して行くこと

ができた。しかし、以下のような演算はスムーズに行うことができない。

(X<<1)+(Y<<1)

上の式ではまずXをアキュムレータに入れて 1ビット左シフトし、その結果をどこかにとっておいて、それから Yを

左シフトして加えなければならない。つまり、結果をとっておく場所が必要であり、これを可能にするのがメモリで

ある。

デバイスとしてのメモリはディジタル回路の時間に紹介するので、ここでは、メモリは単純に図 1に示す表と考

えて欲しい。この表は、各行に番号（アドレス）が振られており、アドレスの値によって、対応する行の中身（デー

タ）を読み出すことができる。この場合、アドレスに 0を設定すれば 1001が、1を設定すれば 1100が読み出され、

Dataoutに出力される。

また、書き込みの場合、入力にデータを置き、アドレスを設定し、書き込み制御信号 we(write enable)を Hにし、

クロックを立ち上げると、そのアドレスにデータが書き込まれる。

この例は超簡単で、幅が 4ビットで深さが 24=16のメモリということになり、全容量は 64bitである。深さが 16=24

であるため、番地を識別するアドレス線も 4本で良い。もちろん、実際のメモリはもっとずっとずっと大きく、アド

レスが 32bit-48bit、データは 32bit-64bitで膨大な情報量を格納する。しかし原理は同じである。アドレスが m bit

あれば、2m行分指定することができ、容量の大きいメモリほど、アドレスの本数も多い。

図 1: メモリの構成

それでは、深さ 28=256、幅 16bitのメモリを想定し、これを前回のデータパスにつないでみよう。この様子を図 2

に示す。

この構成では、データパスの入力にはメモリの出力、アキュムレータ出力にはメモリの入力が接続されている。す

なわち、データを外から与えてやる代わりにメモリ中にあらかじめ格納してデータに対して演算を行い、途中結果も

メモリに格納することができる。外部から与えるのは ALUの comとメモリのアドレスおよび weである。

例えば Xが 0番地、Yが 1番地に格納されている場合、先の例は以下の値をクロックに同期して順に与えれば計算

できる。

we com Address

0 001 00000000 : 0番地 (X)を読み出し、ALUはスルー (LD)

図 2: メモリ付きデータパス

0 100 00000000 : 1bit 左シフト (Addressはここでは何でも良い)

1 000 00000010 : 2番地に X<<1を格納 (ST)

0 001 00000001 : 1番地 (Y)を読み出し、ALUはスルー (LD)

0 100 00000000 : 1bit 左シフト (Addressはここでは何でも良い)

0 110 00000010 : 2番地の値 (X<<1)と加算

1 000 00000011 : 3番地に答えを格納 (ST)

ここでは 2番地を途中結果を格納する場所として使い、最終結果は 3番地に格納している。ここで、weを 1、com

を 000とすると、アキュムレータの内容が Addressに示す番地に格納される。この操作を Store（ストア）と呼ぶ。

Storeは Loadの逆の操作であり、この言い方は様々なコンピュータで広く用いられる。

ここで、weと comをくっつけてしまって 4ビットとし、命令コード (op-code)と呼ぶことにする。すると上の操

作は以下のように記述することができる。

opcode Address

0001 00000000 : LD 0

0100 00000000 : SL

1000 00000010 : ST 2

0001 00000001 : LD 1

0100 00000000 : SL

0110 00000010 : ADD 2

1000 00000011 : ST 3

ここで、opcodeに対応する ALUの操作および LD/STを、対象とするアドレスと共に記述してやると、操作を見

やすく示すことができる。これがコンピュータの命令 (Instruction)と、そのアセンブリ表記の第一歩である。

命令メモリの取り付け

上のデータパスでは、計算するデータはデータメモリにあらかじめ入れておけば良いのに、命令は外からクロック

に同期して与えてやらなければならず、これは大変である。そこで、データ同様、命令もあらかじめメモリに入れて

おき、これを順番に読み出してデータパスに与えるようにすれば便利である。この構成を図 3 に示す。

ここでは、深さ 28=256、幅 12bitの命令メモリを想定し、ここに命令を入れておく。これを順に読み出すためのし

かけとして、PC(Program Counter:プログラムカウンタ)を用意する。PCはスタート時に 0に初期化しておき、ク

ロックが立ち上がる毎に 1ずつ増やしてやる。そうすると、命令メモリに格納されている命令が順にデータパスに与

えられて、所定の処理が順に実行されていく。

読み出された来た命令の opcodeに対応する部分は最上位ビットが 1の場合は、ST命令であることを示す。この際、

we=1としてデータメモリに accumの値を書き込む。また、ST命令の時は accumに ALUの計算結果を書き込まな

い。このため、accumに accsetというセット入力を与え、これが 1の時のみ、結果を書き込むようにする。1

図 3: 命令メモリ付きデータパス

これが、プログラム格納型計算機の第一歩である。

メモリ付きデーパスの verilog記述

メモリはレジスタの配列として考えて、以下のように reg文で宣言する。

reg [MSB:LSB] 名前 [深さ]

例えば、深さ DEPTHでデータ幅 DATA Wのメモリ dmemを宣言するには

reg [‘DATA_W-1:0] dmem[‘DEPTH-1:0]

とする。それでは、図 3の verilog記述を説明しよう。

ここでは、図の構造がほとんどそのままの形で記述されている。命令メモリから読み出された来た命令は、opcode

とデータメモリに対するアドレス (daddr)に分離される。この opcodeが ST命令のときに書き込みが行われるよう

にしている。逆に accumには、ST命令以外で aluで演算された値が格納されるようになっている。pcはリセット時

に 0になり、後は順に 1ずつ増えて行く。すなわち、0番地から順に命令が実行される。

‘include "def.h"

module datapath2(

input clk, input rst_n,

output [‘DATA_W-1:0] accout);

reg [‘DATA_W-1:0] accum;

1com の下がすべて 0ならば A入力がスルーされて yから accum に書かれるので実は現時点では、この accset は必要ない。しかし後でどの
みち必要になるので付けておくことにする。

reg [‘ADDR_W-1:0] pc;

reg [‘DATA_W-1:0] dmem [‘DEPTH-1:0];

reg [‘INST_W-1:0] imem [‘DEPTH-1:0];

wire [‘DATA_W-1:0] alu_y;

wire [‘DATA_W-1:0] ddatain;

wire [‘OPCODE_W-1:0] opcode;

wire [‘INST_W-1:0] idatain;

wire [‘ADDR_W-1:0] daddr;

wire we;

assign idatain = imem[pc];

assign {opcode, daddr} = idatain;

assign ddatain = dmem[daddr];

assign we = (opcode == ‘OP_ST);

always @(posedge clk)

if(we) dmem[daddr] <= accum;

assign accout = accum;

alu alu_1(.a(accum), .b(ddatain), .s(opcode[‘SEL_W-1:0]), .y(alu_y));

always @(posedge clk or negedge rst_n)

begin

if(!rst_n) pc <= 0;

else pc <= pc + 1;

end

always @(posedge clk or negedge rst_n)

begin

if(!rst_n) accum <= 0;

else if(opcode !=‘OP_ST) accum <= alu_y;

end

endmodule

では次に、これをテストするテストベンチ test.vを解説しよう。

‘timescale 1ns/1ps

‘include "def.h"

module test;

parameter STEP = 10;

reg clk, rst_n;

wire [‘DATA_W-1:0] accout;

always #(STEP/2) begin

clk <= ~clk;

end

datapath2 dp2(.clk(clk), .rst_n(rst_n), .accout(accout));

initial begin

$dumpfile("datapath2.vcd");

$dumpvars(0,test);

$readmemh("dmem.dat", dp2.dmem);

$readmemb("imem.dat", dp2.imem);

clk <= ‘DISABLE;

rst_n <= ‘ENABLE_N;

#(STEP*1/4)

#STEP

rst_n <= ‘DISABLE_N;

#(STEP*12)

$finish;

end

always @(negedge clk) begin

$display("pc:%h inst:%h acc:%h", dp2.pc, dp2.idatain, dp2.accum);

$display("dmem:%h %h %h %h", dp2.dmem[0], dp2.dmem[1], dp2.dmem[2], dp2.dmem[3]);

end

endmodule

ここでは、module datapath2に dp2という実体名を付け、以下の文で命令メモリとデータメモリの初期設定を行っ

ている。

$readmemh("dmem.dat", dp2.dmem);

$readmemb("imem.dat", dp2.imem);

ここで readmemhは 16進数で記述すること、readmembは 2進数で記述されたファイルから値を読み込んで、それ

ぞれ dp2中の dmem, imemに値を設定することを示す。

web上のファイルでは imem.datは

0001_00000000 // LD 0

0111_00000011 // SUB 3

1000_00000011 // ST 3

0001_00000001 // LD 1

0110_00000010 // ADD 2

0011_00000011 // OR 3

1000_00000011 // ST 3

になっている。opcodeとアドレスを区切るのに が使われているが、これは無視される。これに対して dmem.dat

は 16進数で、

0009

0005

0003

0002

となっている。実行して、結果を確認しよう。accumに答えが入るのは命令が実行された次のクロックである点に

注意されたい。

演習 4:

X,Y,W,Zが 0,1,2,3番地に格納されているとする。(X-Z) AND (Y-W)を演算する命令コードを書き、実行して答

えを確認せよ。なお、答えは accum上に出た時点で終わりにして良い。提出物は変更した imem.datである。

